
Introduction

Effective graphics are indispensable in modern statistical analysis. They reveal patterns, differences, and
uncertainty that are not readily apparent in tabular output. Graphics provoke questions that stimulate deeper
investigation, and they add visual clarity and rich content to reports and presentations. Statistical graphs are
produced by ODS Graphics, which is an extension of ODS (the Output Delivery System). ODS manages
procedure output (including both tables and graphs) and lets you display it in a variety of destinations, such as
HTML and RTF. With ODS Graphics, statistical procedures produce graphs as automatically as they produce
tables, and graphs are integrated with tables in the ODS output. ODS Graphics is available in procedures

in SAS/STAT, Base SAS, SAS/ETS, SAS/QC, and other SAS products (see the section “Procedures That
Support ODS Graphics” on page 617). ODS Graphics is automatically provided with Base SAS software.

ODS Graphics might or might not be enabled by default, depending on your operating system, whether
you are in the SAS windowing environment, your registry, your system options, and your con�guration �le
settings. For more information about default settings and enabling and disabling ODS Graphics, see the
section “Enabling and Disabling ODS Graphics” on page 609.

You can enable ODS Graphics by specifying the following statement:

ods graphics on;

When ODS Graphics is enabled, procedures that support ODS Graphics create the appropriate graphs, either
by default or when you specify procedure options to request speci�c graphs. These options are documented
in the “Syntax” section of each procedure chapter, and the “Details” section of each chapter provides an
“ODS Graphics” subsection that lists the available graphs. Once ODS Graphics is enabled, it stays enabled
for the duration of your SAS session unless you disable it.

You can disable ODS Graphics by specifying the following statement:

ods graphics off;

You might consider disabling ODS Graphics if your goal is solely to produce computational results. Often,
you can enable ODS Graphics and then leave it enabled. Throughout this chapter, ODS Graphics is enabled
only once per section.

Chapter Reading Guide

This chapter provides a basic introduction to ODS Graphics along with more detailed information. The
following list provides a guide to reading this chapter:

� If you want to see a few of the many graphs that statistical procedures produce by using ODS Graphics,
see the section “Getting Started with ODS Statistical Graphics” on page 592.

� If you are using ODS Graphics for the �rst time, read the section “A Primer on ODS Statistical
Graphics” on page 608, which provides the minimum information that you need to get started.

� If you need to create plots of raw data or your own customized plots of statistical results, see the
section “Statistical Graphics Procedures” on page 699, which describes SAS procedures that use ODS
Graphics.

� If you need information about specialized topics such as accessing your graphs, making changes to
your graphs, and working with ODS styles, see the section “Syntax” on page 618 and the sections
“Examples of ODS Statistical Graphics” on page 708, “Selecting and Viewing Graphs” on page 624,
“Graphic Image Files” on page 630, “Graph Size and Resolution” on page 636, “ODS Graphics Editor”
on page 637, “The Default Template Stores and the Template Search Path” on page 641, “ODS Styles”
on page 643, and “Examples of ODS Statistical Graphics” on page 708.

If you are unfamiliar with ODS, see Chapter 20, “Using the Output Delivery System,” for an introduction.
For complete documentation about the Output Delivery System, see theSAS Output Delivery System: User's
Guide. For an introduction to graph template modi�cation, see Chapter 22, “ODS Graphics Template
Modi�cation.” For more information about modifying the Kaplan and Meier (1958) plot in PROC LIFETEST,

see Chapter 23, “Customizing the Kaplan-Meier Survival Plot.” For an introduction to ODS Graphics, ODS
styles, the Graph Template Language, the style template language, the statistical graphics procedures, and
graph template modi�cation, see Kuhfeld (2010). For complete documentation about ODS graph templates,
see theSAS Graph Template Language: User's Guideand theSAS Graph Template Language: Reference.
For complete documentation about the ODS Graphics Editor, see theSAS 9.4 ODS Graphics Editor: User's
Guide. Also see theSAS ODS Graphics: Procedures Guidefor information about the statistical graphics
procedures.

Assumptions about ODS Defaults in This Chapter

Default settings such as destinations and whether or not ODS Graphics is enabled vary depending on your
operating system, registry settings, con�guration �le settings, and system options and whether you are using
the SAS windowing environment or batch mode. For this reason, this chapter makes no assumptions about
these defaults. Instead, destinations are often explicitly closed without assuming which destination (usually
LISTING or HTML) is open, destinations are explicitly opened when needed, and ODS Graphics is explicitly
enabled and disabled as needed. In some examples, when all destinations are closed, the LISTING destination
is opened at the end of the step so that some destination is available for subsequent output. If you know
the defaults for your environment, you do not need to use many of the ODS statements that are used in this
chapter.

Getting Started with ODS Statistical Graphics

This section provides examples that illustrate the most basic uses of ODS Graphics by showing a few of the
many plots that are produced by statistical procedures.

Default Plots for Simple Linear Regression with PROC REG

This example is based on the section “Getting Started: REG Procedure” on page 7836 in Chapter 97, “The
REG Procedure.” TheClass data set that this example uses is available in theSashelp library. The following
statements use PROC REG to �t a simple linear regression model in whichWeight is the response variable
andHeight is the independent variable:

ods graphics on;

proc reg data=sashelp.class;
model Weight = Height;

run; quit;

The ODS GRAPHICS ON statement requests ODS Graphics in addition to the usual tabular output. The
statement ODS GRAPHICS OFF is not used here, but it can be speci�ed to disable ODS Graphics.

The graphical output consists of a �t diagnostics panel, a residual plot, and a �t plot. These plots are integrated
with the tabular output and are shown in Figure 21.1, Figure 21.2, and Figure 21.3, respectively.

The results are displayed in the HTMLBLUE style. ODS styles control the colors and general appearance
of all graphs and tables, and the SAS System provides several styles that are recommended for use with
statistical graphics. The default style that you see when you run SAS depends on the ODS destination, system
options, and SAS registry settings. For more information about styles, see the section “ODS Styles” on
page 610 and the section “ODS Styles” on page 643.

Figure 21.1 Fit Diagnostics Panel

Figure 21.2 Residual Plot

Figure 21.3 Fit Plot

Survival Estimate Plot with PROC LIFETEST

This example is taken from Example 70.2 in Chapter 70, “The LIFETEST Procedure.” It shows how to
construct a product-limit survival estimate plot. Both the ODS GRAPHICS statement and procedure options
are used to request the plot. This examples uses the bone marrow transplant data set, which is available from
theSashelp library. The data set contains disease-free times for three risk categories.

The following statements use PROC LIFETEST to compute the product-limit estimate of the survivor function
for each risk category:

proc lifetest data=sashelp.BMT plots=survival(cb=hw test);
time T * Status(0);
strata Group / test=logrank;

run;

ODS Graphics was enabled in a previous step, and the PLOTS=SURVIVAL option requests a plot of the
estimated survival curves. The CB=HW suboption requests Hall-Wellner con�dence bands, and the TEST
suboption displays thep-value for the log-rank test in a plot inset. For more information about modifying
the Kaplan and Meier (1958) plot in PROC LIFETEST, see Chapter 23, “Customizing the Kaplan-Meier
Survival Plot.”

Figure 21.4 displays the plot. Patients in the AML-Low Risk group are disease-free longer than those in the
ALL group, who in turn fare better than those in the AML-High Risk group.

Figure 21.4 Survival Plot

Contour and Surface Plots with PROC KDE

This example is taken from the section “Getting Started: KDE Procedure” on page 4874 in Chapter 66, “The
KDE Procedure.” Here, in addition to the ODS GRAPHICS statement, procedure options are used to request
plots. The following statements simulate 1,000 observations from a bivariate normal density that has means
(0,0), variances (10,10), and covariance 9:

data bivnormal;
do i = 1 to 1000;

z1 = rannor(104);
z2 = rannor(104);
z3 = rannor(104);
x = 3 * z1+z2;
y = 3 * z1+z3;
output;

end;
run;

The following statements request a bivariate kernel density estimate for the variablesx andy:

proc kde data=bivnormal;
bivar x y / plots=contour surface;

run;

The PLOTS= option requests a contour plot and a surface plot of the estimate (displayed in Figure 21.5 and
Figure 21.6, respectively). For more information about the graphs available in PROC KDE, see the section
“ODS Graphics” on page 4893 in Chapter 66, “The KDE Procedure.”

Figure 21.5 Contour Plot of Estimated Density

Figure 21.6 Surface Plot of Estimated Density

Contour Plots with PROC KRIGE2D

This example is taken from Example 67.2 in Chapter 67, “The KRIGE2D Procedure.” The coal seam
thickness data set is available from theSashelp library. The following statements create a SAS data set that
contains a copy of these data along with some arti�cially added missing data:

data thick;
set sashelp.thick;
if _n_ in (41, 42, 73) then thick = .;

run;

The following statements run PROC KRIGE2D:

proc krige2d data=thick outest=predictions
plots=(observ(showmissing)

pred(fill=pred line=pred obs=linegrad)
pred(fill=se line=se obs=linegrad));

coordinates xc=East yc=North;
predict var=Thick r=60;
model scale=7.2881 range=30.6239 form=gauss;
grid x=0 to 100 by 2.5 y=0 to 100 by 2.5;

run;

The PLOTS=OBSERV(SHOWMISSING) option produces a scatter plot of the data along with the locations
of any missing data. The PLOTS=PRED option produces maps of the kriging predictions and standard errors.
Two instances of the PLOTS=PRED option are speci�ed along with suboptions that customize the plots. The
results are shown in Figure 21.7.

Figure 21.7 Spatial Distribution

Figure 21.7 continued

Partial Least Squares Plots with PROC PLS

This example is taken from the section “Getting Started: PLS Procedure” on page 7053 in Chapter 88, “The
PLS Procedure.” The following statements create a SAS data set that contains measurements of biological
activity in the Baltic Sea:

data Sample;
input obsnam $ v1-v27 ls ha dt @@;
datalines;

EM1 2766 2610 3306 3630 3600 3438 3213 3051 2907 2844 2796
2787 2760 2754 2670 2520 2310 2100 1917 1755 1602 1467
1353 1260 1167 1101 1017 3.0110 0.0000 0.00

EM2 1492 1419 1369 1158 958 887 905 929 920 887 800

... more lines ...

;

The following statements run PROC PLS:

proc pls data=sample cv=split cvtest(seed=104);
model ls ha dt = v1-v27;

run;

By default, the procedure produces a plot for the cross validation analysis and a correlation loading plot (see
Figure 21.8).

Figure 21.8 Partial Least Squares

Figure 21.8 continued

Box-Cox Transformation Plot with PROC TRANSREG

This example is taken from Example 117.2 in Chapter 117, “The TRANSREG Procedure.” The following
statements create a SAS data set that contains failure times for yarn:

proc format;
value a -1 = 8 0 = 9 1 = 10;
value l -1 = 250 0 = 300 1 = 350;
value o -1 = 40 0 = 45 1 = 50;

run;

data yarn;
input Fail Amplitude Length Load @@;
format amplitude a. length l. load o.;
label fail = �Time in Cycles until Failure�;
datalines;

674 -1 -1 -1 370 -1 -1 0 292 -1 -1 1 338 0 -1 -1
266 0 -1 0 210 0 -1 1 170 1 -1 -1 118 1 -1 0

... more lines ...

;

The following statements run PROC TRANSREG:

proc transreg data=yarn;
model BoxCox(fail / convenient lambda=-2 to 2 by 0.05) =

qpoint(length amplitude load);
run;

The log-likelihood plot in Figure 21.9 suggests a Box-Cox transformation where� D 0.

Figure 21.9 Box-Cox “Signi�cant Effects”

LS-Means Diffogram with PROC GLIMMIX

This example is taken from the section “Graphics for LS-Mean Comparisons” on page 3394 in Chapter 45,
“The GLIMMIX Procedure.” The following statements create a SAS data set that contains measurements
from an experiment that investigates how snapdragons grow in various soils:

data plants;
input Type $ @;
do Block = 1 to 3;

input StemLength @@;
output;

end;
datalines;

Clarion 32.7 32.3 31.5 Clinton 32.1 29.7 29.1 Knox 35.7 35.9 33.1
ONeill 36.0 34.2 31.2 Compost 31.8 28.0 29.2 Wabash 38.2 37.8 31.9
Webster 32.5 31.1 29.7
;

The following statements run PROC GLIMMIX:

proc glimmix data=plants order=data plots=diffogram;
class Block Type;
model StemLength = Block Type;
lsmeans Type;

run;

The PLOTS=DIFFOGRAM option produces a diffogram, shown in Figure 21.10, that displays all the pairwise
least squares mean differences and indicates which are signi�cant.

Figure 21.10 LS-Means Diffogram

Principal Component Analysis Plots with PROC PRINCOMP

This example is taken from Example 91.3 in Chapter 91, “The PRINCOMP Procedure.” The following
statements create a SAS data set that contains job performance ratings of police of�cers:

options validvarname=any;

data Jobratings;
input (�Communication Skills�n

�Problem Solving�n
�Learning Ability�n
�Judgment Under Pressure�n
�Observational Skills�n

�Willingness to Confront Problems�n
�Interest in People�n
�Interpersonal Sensitivity�n
�Desire for Self-Improvement�n
�Appearance�n
�Dependability�n
�Physical Ability�n
�Integrity�n
�Overall Rating�n) (1.);

datalines;
26838853879867
74758876857667

... more lines ...

;

The following statements run PROC PRINCOMP:

proc princomp data=Jobratings(drop=�Overall Rating�n) n=2
plots=(Matrix PatternProfile);

run;

The plots are requested by the PLOTS=(MATRIX PATTERNPROFILE) option. The results, shown in
Figure 21.11, contain the default scree and variance-explained plots, along with a scatter plot matrix of
component scores and a pattern pro�le plot.

Figure 21.11 Principal Component Analysis

Figure 21.11 continued

Figure 21.11 continued

Grouped Scatter Plot with PROC SGPLOT

This example is taken from Example 35.1 in Chapter 35, “The DISCRIM Procedure.” It uses the Fisher iris
data set, which is available from theSashelp library.

The following statements run PROC SGPLOT to make a scatter plot, grouped by iris species:

proc sgplot data=sashelp.iris;
title �Fisher (1936) Iris Data�;
scatter x=petallength y=petalwidth / group=species;

run;

The results are shown in Figure 21.12.

Figure 21.12 Iris Data

For more information about PROC SGPLOT (statistical graphics plot) and other SG procedures, see the
section “Statistical Graphics Procedures” on page 699 and theSAS ODS Graphics: Procedures Guide. You
do not need to enable ODS Graphics in order to use SG procedures (because making plots by using ODS
Graphics is their sole function). However, you can use the ODS GRAPHICS statement to change options, as
in the following example:

ods graphics on / attrpriority=none;

proc sgplot data=sashelp.iris;
title �Fisher (1936) Iris Data�;
styleattrs datasymbols=(circlefilled squarefilled starfilled);
scatter x=petallength y=petalwidth / group=species markerattrs=(size=5px);

run;

ods graphics / reset;

The STYLEATTRS statement speci�es the symbols for the three groups of observations. The results are
shown in Figure 21.13. The groups in Figure 21.13 are distinguished by varying colors and symbols. In
contrast, the groups in Figure 21.12 are distinguished only by colors. The ATTRPRIORITY=NONE option
in the ODS GRAPHICS statement ensures that the second and third symbols are used. The HTMLBLUE
style is an ATTRPRIORITY=“Color” style, so by default, the symbol and line style attributes are not used to
distinguish groups. The ATTRPRIORITY=NONE option in the ODS GRAPHICS statement overrides the
ATTRPRIORITY=“Color” option in the style, so that all three symbols are used. For more information, see
the sections “Attribute Priorities” on page 646 and “Overriding How Groups Are Distinguished” on page 647.

Figure 21.13 Iris Data Controlling the Symbols

A Primer on ODS Statistical Graphics

You can enable ODS Graphics by specifying the following statement:

ods graphics on;

ODS Graphics remains enabled for all procedure steps until you disable it by submitting the following
statement:

ods graphics off;

When ODS Graphics is enabled, creating graphical output with procedures is as simple as creating tabular
output. For more information about enabling and disabling ODS Graphics, see the section “Enabling
and Disabling ODS Graphics” on page 609. For more information about the most commonly used ODS
GRAPHICS statement options, see the section “Syntax” on page 618.

You can control your output by using the following ODS components:

� ODS destination statements (such as ODS HTML or ODS RTF), which specify where you want your
graphs to be displayed. For an example of HTML output, see Figure 21.23. For a list of the supported
destinations, see the section “ODS Destination Statements” on page 621. For more information about
the most commonly used ODS destination statement options, see the section “Syntax” on page 618 .

� ODS SELECT and ODS EXCLUDE statements, which select and exclude graphs from your output.
See the section “Selecting and Excluding Graphs” on page 628 for an example of how to select graphs.

� ODS OUTPUT statements, which create SAS data sets from the data object that is used to make the
plot. See the section “Specifying an ODS Destination for Graphics” on page 624 for an example.

� procedure options, which specify what graphs to create. For each procedure, these options are described
in the “Syntax” section of the procedure chapter. Usually, you use the PLOTS= option to control all
graphs. The available graphs are listed in the “ODS Graphics” section, which is found in the “Details”
section of each procedure chapter. Many graphs are produced by default.

� ODS style templates, which control the general appearance and consistency of all graphs and tables.
You can modify the styles that SAS provides and override style information in several ways. For more
information about styles, see the sections “ODS Styles” on page 610 and “ODS Styles” on page 643.

� ODS graph templates, which modify the layout and details of each graph. For more information
about graph templates, see the section “Graph Templates” on page 722 in Chapter 22, “ODS Graphics
Template Modi�cation.”

NOTE : SAS provides a default template for each graph, so you do not need to know anything
about templates in order to create statistical graphics.

You can also access individual graphs, control the resolution and size of graphs, and modify your graphs (as
explained in the sections beginning with “Selecting and Viewing Graphs” on page 624). Alternatively, you
can use special statistical graphics procedures to create custom graphs directly (see the section “Statistical
Graphics Procedures” on page 699).

Enabling and Disabling ODS Graphics

You can enable ODS Graphics by specifying the following statement:

ods graphics on;

ODS Graphics remains enabled for all procedure steps until you disable it by submitting the following
statement:

ods graphics off;

ODS Graphics might or might not be enabled by default. This depends on a number of factors. ODS
Graphics is usually enabled by default in the SAS windowing environment; ODS Graphics is usually disabled

by default when you invoke SAS in other ways. However, you can change these defaults in a number of
ways. You can enable or disable ODS Graphics by default in anautoexec.sas �le, a con�guration �le such
asSASV9.CFG, or the SAS registry. You can change the default in the SAS windowing environment by
selectingTools I Options I Preferencesfrom the main SAS window. Then on theResultstab, select the
Use ODS Graphicscheck box to enable ODS Graphics by default, or clear the check box to disable ODS
Graphics by default. You can also change the default output destination (HTML or LISTING) on theResults
tab. For more information about default ODS Graphics settings and default destinations, see the section
“HTML Output in the SAS Windowing Environment” on page 522 in Chapter 20, “Using the Output Delivery
System.”

When ODS Graphics is enabled, procedures that support ODS Graphics create graphs, either by default or
when you specify procedure options for requesting speci�c graphs. Often, you can leave ODS Graphics
enabled for the duration of your SAS session. However, you might consider disabling ODS Graphics if your
goal is solely to produce computational results, particularly for large data sets or with many BY groups.

ODS Styles

ODS styles control the overall appearance of graphs and tables. They specify colors, fonts, line styles, symbol
markers, and other attributes of graph elements. There are two types of ODS styles:

� ATTRPRIORITY=“Color” style, which distinguishes groups of observations by color changes and not
by line style or symbol changes.1 ATTRPRIORITY=“Color” styles include HTMLBLUE, PEARL,
PEARLJ, and SAPPHIRE. If you want to control the markers or lines that are displayed for groups of
observations when using an ATTRPRIORITY=“Color” style, be sure to �rst specify the ATTRPRI-
ORITY=NONE option in the ODS GRAPHICS statement or switch to an ATTRPRIORITY=“None”
style. For more information, see the sections “Attribute Priorities” on page 646 and “Overriding How
Groups Are Distinguished” on page 647.

� ATTRPRIORITY=“None” style, which distinguishes groups of observations by simultaneous color,
marker, and line changes. Most ODS styles are ATTRPRIORITY=“None” styles. They are com-
promise styles in the sense that some graph elements are intentionally overdistinguished to facilitate
black-and-white printing. For example, �t lines that correspond to different classi�cation levels are
distinguished by both colors and line patterns. You can use the ATTRPRIORITY=“Color” styles (such
as HTMLBLUE, PEARL, PEARLJ, and SAPPHIRE) when you want groups to be distinguished only
by color.

Although you can use any ODS style, only a few styles are usually used with ODS Graphics. They are
described in Table 21.1.

1More precisely, an ATTRPRIORITY=“Color” style such as HTMLBLUE distinguishes the �rst 12 groups of observations only
by color. Markers and lines change for groups 13–24 and then again for groups 25–36. Figure 21.54 shows how colors, markers, and
line styles change in the HTMLBLUE style, and Figure 21.53 shows how these change in most other ODS styles.

Table 21.1 ODS Styles Most Often Used with ODS Graphics

Style
Recommended
Destinations

Attribute
Priority Description

ANALYSIS HTML None A color style, with sans serif fonts, whose dominant
colors are yellow, green, and tan. See Figure 21.26.

DEFAULT HTML None A color style, with bold sans serif fonts, whose dominant
colors are gray, blue, and white. See Figure 21.22.

HTMLBLUE HTML Color X A color style, with sans serif fonts, whose dominant
colors are shades of blue. See Figure 21.23. Default for
HTML destination and SAS/STAT documentation.

HTMLBLUECML HTML NoneX An ATTRPRIORITY=“None” version of HTMLBLUE.
See Figure 21.24.

JOURNAL,
JOURNAL1A

PDF, PS, RTF,
PRINTER

NoneX A black-and-white style with sans serif fonts and �lled
areas. See Figure 21.27, Figure 21.32, and Figure 21.33.

JOURNAL2,
JOURNAL2A

PDF, PS, RTF,
PRINTER

NoneX A black-and-white style, similar to JOURNAL but with
empty areas. Grouped bar charts use crosshatching to
show groups. See Figure 21.34 and Figure 21.35.

JOURNAL3,
JOURNAL3A

PDF, PS, RTF,
PRINTER

NoneX A black-and-white style, similar to JOURNAL2 but with
a mix of �lled areas and crosshatching in grouped bar
charts. See Figure 21.36 and Figure 21.37.

LISTING HTML,
LISTING

None A color style, similar to DEFAULT but with a white
background. See Figure 21.28. Default for the LISTING
destination.

PEARL PDF, PS, RTF,
PRINTER

ColorX A color style, with sans serif fonts and a white back-
ground, whose dominant colors are shades of blue.
See Figure 21.30 and Figure 21.38. Default for PDF
destination.

PEARLJ PDF, PS, RTF,
PRINTER

ColorX A color style, with sans serif fonts and a white back-
ground, whose dominant colors are shades of blue. See
Figure 21.39. Default for PDF tables in SAS/STAT
documentation.

RTF RTF None A color style, with serif (Times Roman) fonts, whose
dominant colors are blue, white, and black. See Fig-
ure 21.29 and Figure 21.40. Default for RTF destination.

SAPPHIRE PDF, PS, RTF,
PRINTER

ColorX A color style, with sans serif fonts, a white background,
and a light blue table heading background, whose dom-
inant colors are shades of blue. See Figure 21.31 and
Figure 21.41.

STATISTICAL HTML None A color style, with sans serif fonts, whose dominant
colors are blue, gray, and white. See Figure 21.25.

X indicates newer styles that are recommended for use with statistical graphics.
JOURNAL# styles differ from JOURNAL#A styles in that the former use italic fonts in table headings.

You specify an ODS style by using the STYLE= option in the ODS destination statement. For example, the
following statement creates RTF output and speci�es the JOURNAL style:

ods rtf style=Journal;

The following statement sets the style for the LISTING destination:

ods listing style=HTMLBlue;

The style that is speci�ed by the STYLE= option in the ODS LISTING statement applies only to graphs.
SAS monospace format is used for tables.

More generally, you can modify the colors, fonts, and other attributes of graph elements in an ODS style by
editing the style template. For more information, see the section “ODS Styles” on page 643 andSAS Output
Delivery System: User's Guide. You can also perform ODS style modi�cations by using the %MODSTYLE
SAS autocall macro. For more information, see the section “ODS Style Template Modi�cation Macro” on
page 685.

ODS Destinations

ODS can send your graphs and tables to a number of different destinations, including RTF (rich text format),
HTML (hypertext markup language), LISTING (the SAS LISTING destination), DOCUMENT (the ODS
document), and PDF (portable document format). You use an ODS statement to open a destination, as in the
following examples:

ods html body=�b.htm�;
ods rtf;
ods listing;
ods document name=MyDoc(write);
ods pdf file="contour.pdf";

You can close destinations individually or all at once, as in the following examples:

ods html close;
ods rtf close;
ods listing close;
ods document close;
ods pdf close;
ods _all_ close;

For most ODS destinations (such as HTML, RTF, and PDF), graphs and tables are integrated in the output,
and you view your output by using an appropriate viewer, such as a web browser for HTML. However,
the LISTING destination is different. If you are using the LISTING destination in the SAS windowing
environment, you view your graphs individually by clicking the graph icons in the Results window, shown in
Figure 21.14. This action invokes a host-dependent graph viewer (for example, Microsoft Photo Editor in
Windows). The graphs that ODS Graphics produces arenot displayed along with traditional graphs in the
Graph window.

Figure 21.14 SAS Results Window

If you are using the SAS windowing environment and you prefer to view integrated output, you should use a
destination such as HTML or RTF. In many cases, HTML is the default destination in the SAS windowing
environment (see the section “HTML Output in the SAS Windowing Environment” on page 522 in Chapter 20,
“Using the Output Delivery System”). You can change destinations in the SAS windowing environment by
selectingTools I Options I Preferencesfrom the main SAS window and then clicking theResultstab.

You can prevent the Output window from appearing by using ODS statements to close the LISTING
destination, as follows:

ods listing close;
ods html;

ODS creates a graph for every open destination. When you open a new destination, you should close all
destinations that you do not need. Closing destinations makes your jobs run faster and use fewer resources,
because fewer tables and graphs are produced.

Accessing Individual Graphs

If you are writing a paper or creating a presentation, you need to access your graphs individually. There are
various ways to do this, depending on the ODS destination. Three particularly useful methods are as follows:

� If you are viewing RTF output, you can simply copy your graphs from the viewer and paste them into
a Microsoft Word document or PowerPoint slide.

� If you are viewing HTML output, you can copy and paste your graphs from the viewer, or you can
right-click the graph and save it to a �le. For more information, see the section “Specifying the Size
and Resolution of Graphs” on page 615.

� You can save your graphs in image �les and then include them in a paper or presentation. For example,
you can save your graphs as PNG �les and include them in an HTML document or in a paper that you
are writing in LATEX.

You can specify the graphics image format and the �lename in the ODS GRAPHICS statement. You can
specify the graph resolution by using the IMAGE_DPI= option in an ODS destination statement. For example,
the following statements, when submitted before a procedure step that produces multiple graphs, save the
graphs in PostScript �les namedmyname.ps, myname1.ps, and so on:

ods _all_ close;
ods latex;
ods graphics on / outputfmt=ps imagename=�myname�;

The following statements save the graphs in PNG �les with a resolution of 300 dots per inch (DPI):

ods _all_ close;
ods html image_DPI=300;
ods graphics on;

For more information about the �le types available with various destinations, how they are named, and how
they are saved, see the section “Image File Types” on page 630. For more information about resolution, see
the section “Specifying the Size and Resolution of Graphs” on page 615. For more information about scalable
vector graphics, see the section “Scalable Vector Graphics” on page 631. If you are using the LISTING
destination and the SAS windowing environment, you can copy from the viewer into a Microsoft Word
document or PowerPoint slide.

Specifying the Size and Resolution of Graphs

Two factors to consider when you are creating graphs for a paper or presentation are the size of the graph
and its resolution. You can specify the size of a graph in the ODS GRAPHICS statement. The following
examples show typical ways to change the size of your graphs:

ods graphics on / width=6in;
ods graphics on / height=4in;
ods graphics on / width=4.5in height=3.5in;

You can change the resolution by specifying the IMAGE_DPI= option in any ODS destination statement, as
in the following example:

ods html image_dpi=300;

The default resolution of graphs that you create by using the HTML and LISTING destinations is 96 DPI,
whereas the default resolution in the RTF destination is 200 DPI. An increase in resolution often improves
the quality of the graphs, but it also increases the size of the image �le. For more information about graph
size and resolution, see the section “Graph Size and Resolution” on page 636.

Modifying Your Graphs

Although ODS Graphics is designed to automate the creation of high-quality statistical graphics, you might
occasionally need to modify your graphs. There are two ways to do this, depending on whether the changes
that you want to make are data-dependent and immediate (for a speci�c graph you are preparing for a paper
or presentation) or persistent (applied to a graph each time you run the procedure). You can make immediate,
ad hoc changes by using the ODS Graphics Editor, which provides a point-and-click interface. You can
make persistent changes by modifying the ODS graph template for a particular plot. (For an introduction to
graph template modi�cation, see Chapter 22, “ODS Graphics Template Modi�cation.”) A graph template is a
program, written in the Graph Template Language (GTL), that speci�es the layout and details of a graph.

NOTE : The SAS System provides a template for each graph that it creates, so you do not need to know
anything about templates in order to create statistical graphics.

You can use the ODS Graphics Editor to customize titles and labels, annotate data points, add text, and
change the properties of graph elements. After you modify your graph, you can save it as a PNG image �le
or an SGE �le; the latter preserves the editing context. You can open SGE �les by using the ODS Graphics
Editor and resume editing.

You can invoke the ODS Graphics Editor in the SAS windowing environment, provided that you have enabled
ODS Graphics to create editable graphs. The steps for doing this are described in the section “ODS Graphics
Editor” on page 637. Also seeSAS 9.4 ODS Graphics Editor: User's Guide.

Figure 21.15 shows the ODS Graphics Editor window for a �t plot that is created by PROC REG. Figure 21.16
shows modi�cations that are made by using tools in the ODS Graphics Editor. The title has been changed,
and the legend has been repositioned.

Figure 21.15 ODS Graphics Editor Invoked to Edit a Fit Plot

Figure 21.16 Point-and-Click Modi�cations Made Using the ODS Graphics Editor

Procedures That Support ODS Graphics

SAS procedures that support ODS Graphics include the following:

SAS/STAT Base SAS SAS/ETS
ADAPTIVEREG MULTTEST CORR ARIMA
ANOVA NLIN FREQ AUTOREG
BCHOICE NPAR1WAY UNIVARIATE COPULA
BOXPLOT ORTHOREG COUNTREG
CALIS PHREG SAS/QC ENTROPY
CLUSTER PLM ANOM ESM
CORRESP PLS CAPABILITY EXPAND
FACTOR POWER CUSUM HPCDM
FMM PRINCOMP MACONTROL HPQLIM
FREQ PRINQUAL MVPDIAGNOSE HPSEVERITY
GAM PROBIT MVPMONITOR MODEL
GAMPL QUANTLIFE MVPMODEL PANEL
GEE QUANTREG PARETO PDLREG
GENMOD QUANTSELECT RAREEVENTS QLIM
GLIMMIX REG RELIABILITY SEVERITY
GLM ROBUSTREG SHEWHART SIMILARITY
GLMPOWER RSREG SSM
GLMSELECT SEQDESIGN Other SYSLIN
HPFMM SEQTEST HPF TIMEDATA
HPSPLIT SIM2D HPFENGINE TIMEID
ICLIFETEST SPP TIMESERIES
ICPHREG STDRATE SAS Risk UCM
IRT SURVEYFREQ Dimensions VARMAX
KDE SURVEYLOGISTIC X12
KRIGE2D SURVEYMEANS
LIFEREG SURVEYPHREG
LIFETEST SURVEYREG
LOESS TPSPLINE
LOGISTIC TRANSREG
MCMC TTEST
MDS VARCLUS
MI VARIOGRAM
MIXED

For information about the speci�c graphs that a particular procedure creates, see the PLOTS= option syntax
and the “ODS Graphics” section in the corresponding procedure chapter. For the SAS/STAT procedures, the
procedure names in the preceding table link to the “ODS Graphics” section.

Procedures That Support ODS Graphics and Traditional Graphics

A number of procedures that support ODS Graphics produced traditional graphics in previous releases
of SAS. These include the UNIVARIATE procedure in Base SAS software; the LIFEREG, LIFETEST,

and REG procedures in SAS/STAT software; and the ANOM, CAPABILITY, CUSUM, MACONTROL,
PARETO, RELIABILITY, and SHEWHART procedures in SAS/QC software. All these procedures continue
to produce traditional graphics, but in some cases they do so only when ODS Graphics is not enabled. For
more information about the interaction between traditional graphics and ODS graphics in other procedures,
see the documentation for those procedures.

Traditional graphs are saved in SAS graphics catalogs and are controlled by the GOPTIONS statement. In
contrast, ODS Graphics produces graphs in standard image �le formats (not graphics catalogs), and their
appearance and layout are controlled by ODS styles and templates.

Syntax

The following sections document some of the most commonly used options in the ODS GRAPHICS statement
(see the section “ODS GRAPHICS Statement” on page 618) and other ODS Graphics statements (see the
section “ODS Destination Statements” on page 621). You can �nd the complete syntax in theSAS Output
Delivery System: User's Guide. In addition, information about the PLOTS= option is provided in the section
“PLOTS= Option” on page 622. All statistical procedures that produce ODS Graphics have a PLOTS= option
that selects graphs and controls some aspects of the graphs.

ODS GRAPHICS Statement

ODS GRAPHICS < OFF | ON > < / options > ;

The ODS GRAPHICS statement enables ODS to create graphs. You can enable ODS Graphics by using
either of the following equivalent statements:

ods graphics on;
ods graphics;

You specify one of these statements before invoking your procedure, as illustrated in the examples beginning
in the section “Default Plots for Simple Linear Regression with PROC REG” on page 592. Any procedure
that supports ODS Graphics then produces graphs, either by default or when you specify procedure options
to request particular graphs.

To disable ODS Graphics, specify the following statement:

ods graphics off;

ODS Graphics might or might not be enabled by default, depending on your operating system, whether
you are in the SAS windowing environment, your registry, your system options, and your con�guration �le
settings. For more information about default settings and enabling and disabling ODS Graphics, see the
section “Enabling and Disabling ODS Graphics” on page 609.

The following is a subset of the options, syntax, and capabilities available in the ODS GRAPHICS statement.
For more information, see theSAS Output Delivery System: User's Guide.

ANTIALIAS=ON | OFF
controls the use of antialiasing to smooth the components of a graph. Without antialiasing, pixels
are simply set or not set. With antialiasing, pixels at the edge of a line or other object are set to

an intermediate color, which makes smoother and more professional-looking graphics. Text that is
displayed in a graph is always antialiased. Antialiasing is very time-consuming for larger graphs, and its
bene�ts decrease as the number of points increases, so it is disabled by default for plots that have many
points. If the number of graph elements exceeds the ANTIALIASMAX= threshold (4,000 by default),
then antialiasing is not used, even if you specify ANTIALIAS=ON. By default, ANTIALIAS=ON.

ANTIALIASMAX= n
speci�es the maximum number of markers or lines to be antialiased before antialiasing is disabled. For
example, if there are more than 4,000 markers and ANTIALIASMAX=4,000 (the default), then no
markers are antialiased.

ATTRPRIORITY=COLOR | NONE
speci�es the priority for cycling group attributes.

COLOR cycles through the lists of colors while holding the marker symbol and line pattern
constant. When all the colors are exhausted, the marker symbol and line style attributes
increment to the next element, and then the colors in the list are repeated.

NONE simultaneously cycles through colors, marker symbols, and line patterns.

You can specify ATTRPRIORITY=COLOR or ATTRPRIORITY=NONE in the ODS GRAPHICS
statement to change the attribute priority for all graphs. You can specify ATTRPRIORITY=COLOR
or ATTRPRIORITY=NONE in the BEGINGRAPH statement in the GTL to change the attribute
priority for individual graphs. The default attribute priority is set by the ATTRPRIORITY=“Color” or
ATTRPRIORITY=“None” option in the CLASS GRAPH statement in each style.

BORDER=ON | OFF
speci�es whether to draw the graph with a border. By default, BORDER=ON.

BYLINE=FOOTNOTE | TITLE | NOBYLINE
speci�es how the BY-group line is displayed in graphs when the analysis contains a BY statement. By
default, a BY line is not displayed. Specify BYLINE=FOOTNOTE (recommended) to display the BY
line as a left-justi�ed graph footnote or BYLINE=TITLE (not recommended) to display the BY line as
a centered graph title. Most graph templates control the placement of the BY line as follows:

if (_BYTITLE_)
entrytitle _BYLINE_ / textattrs=GraphValueText;

else
if (_BYFOOTNOTE_)

entryfootnote halign=left _BYLINE_;
endif;

endif;

You can modify the graph template if you want the BY line to be displayed in some other way. Because
most graphs have titles and few graphs have footnotes, the BY line looks better when it is displayed as
a footnote.

HEIGHT=dimension
speci�es the height of the graph. By default, HEIGHT=480PX (480 pixels). You can also specify
height in inches (for example, HEIGHT=5IN) or centimeters (for example, HEIGHT=12CM).

IMAGEMAP=ON | OFF
controls tooltip generation in the HTML destination. By default, IMAGEMAP=OFF, which means
that no tooltips are generated. Tooltips are text boxes that appear in HTML output when you rest your
mouse pointer over a part of the plot (see Example 21.1).

IMAGENAME=< base-�le-name >
speci�es the base image �lename. The default is the name of the output object. You can determine the
name of the output object by using the ODS TRACE statement (see the section “Determining Graph
Names and Labels” on page 626). The base image �lename should not include an extension. ODS
automatically adds the increment value and the appropriate extension (which is speci�c to the output
destination). See the section “Specifying Base Filenames” on page 632 for an example.

LABELMAX= n
speci�es the maximum number of labeled areas before labeling is disabled. For example, if LABEL-
MAX=50, and there are more than 50 points that have labels, then no points are labeled. By default,
LABELMAX=200.

MAXLEGENDAREA= n
speci�es the maximum percentage of the overall graph area that a legend can occupy. By default,
MAXLEGENDAREA=20. Larger legends are dropped from the display.

OUTPUTFMT=< image-�le-type | STATIC >
speci�es the image format for graphs. The OUTPUTFMT= option was previously named the IM-
AGEFMT= option. By default, OUTPUTFMT=STATIC, and ODS dynamically uses the best-quality
static image format for the active output destination. The available image formats include BMP (Mi-
crosoft Windows device-independent bitmap), DIB (Microsoft Windows device-independent bitmap),
EMF (Microsoft NT enhanced meta�le), EPSI (Adobe encapsulated PostScript interchange), GIF
(graphic interchange format), JFIF (JPEG �le interchange format), JPEG (Joint Photographic Experts
Group format), PBM (portable bitmap), PCD (Photo CD), PCL (Printer Command Language), PDF
(portable document format), PICT (QuickDraw picture format), PNG (Portable Network Graphics),
PS (PostScript image �le format), SVG (Scalable Vector Graphics format), TIFF (Tagged Image File
Format), WMF (Microsoft Windows Meta�le format), XBM (X Bitmap), and XPM (X-Windows
Pixelmap). If the speci�ed image format is not valid for the active output destination, the device is
automatically remapped to the default image format.

RESET< =option >
resets one or more ODS GRAPHICS options to their default settings. RESET and RESET=ALL
are equivalent. If you want to reset more than one option, but not all the options, then you must
specify RESET= separately for each option that you reset (for example,ods graphics on /
reset=antialias reset=index; ). The RESET=options include the following:

ALL resets all resettable options to their defaults.
ANTIALIAS resets the ANTIALIAS= option to its default.
ANTIALIASMAX resets the ANTIALIASMAX= option to its default.
BORDER resets the BORDER= option to its default.
INDEX resets the index counter that is appended to static image �les.
HEIGHT resets the HEIGHT= option to its default.
IMAGEMAP resets the IMAGEMAP= option to its default.
LABELMAX resets the LABELMAX= option to its default.

SCALE resets the SCALE= option to its default.
TIPMAX resets the TIPMAX= option to its default.
WIDTH resets the WIDTH= option to its default.

SCALE=ON | OFF
speci�es whether the fonts and symbol markers are scaled proportionally to the size of the graph. By
default, SCALE=ON. For examples, see Figure 21.73 and Figure 21.74.

SCALEMARKERS=ON | OFF
speci�es whether markers are scaled in nested layouts. By default, SCALEMARKERS=ON.

TIPMAX=n
speci�es the maximum number of distinct tooltips that are permitted before tooltips are disabled.
Tooltips are text boxes that appear when you rest your mouse pointer over part of the plot. For example,
if TIPMAX=400, and there are more than 400 points in a scatter plot, then no tooltips appear. By
default, TIPMAX=500.

WIDTH=dimension
speci�es the width of the graph. By default, WIDTH=640PX (640 pixels). You can also specify width
in inches (for example, WIDTH=5IN) or centimeters (for example, WIDTH=12CM).

ODS Destination Statements

ODS has a number of statements that control the destination of ODS output. The ODS destination statements
that are most commonly used in ODS Graphics are ODS DOCUMENT, ODS HTML, ODS LISTING, ODS
PCL, ODS PDF, ODS PS, and ODS RTF. Specifying a statement opens a destination, unless the CLOSE
option is speci�ed. Each of the following statements opens an ODS destination:

ods html;
ods rtf;
ods html image_dpi=300;
ods listing style=HTMLBlue;

Each of the following statements closes an ODS destination:

ods html close;
ods rtf close;
ods listing close;

The following statement closes all open destinations:

ods _all_ close;

The following two options are commonly used in ODS destination statements to control aspects of ODS
Graphics:

IMAGE_DPI=dpi
speci�es the image resolution of graphical output, measured in number of dots per inch (DPI). The
default varies depending on the destination. For example, the default is 96 for HTML and 200 for RTF.

STYLE=style-name
speci�es the output style. Commonly used styles include HTMLBLUE, HTMLBLUECML, PEARL,
PEARLJ, SAPPHIRE, DEFAULT, LISTING, STATISTICAL, JOURNAL, JOURNAL2, JOURNAL3,
RTF, and ANALYSIS.

Other options provide ways to control the �les that you create. For example, the following statement opens
the HTML destination:

ods html body=�b.html� contents=�c.html� frame=�a.html�;

This statement also writes the body of the output to the �leb.html, the table of contents to the �lec.html,
and an overall frame that contains both the contents and the output to the �lea.html. Alternatively, you can
specify FILE= instead of BODY=.

If you are using a destination for which individual graphs are created (for example, LISTING or HTML), you
can use the GPATH= option to specify the directory where your graphics �les are saved, as in the following
example:

ods html gpath="C:\figures";

For more information about individual image �les and options speci�ed in the ODS destination statements,
see the sections “Image File Types” on page 630, “Saving Graphic Image Files” on page 633, “LISTING
Destination” on page 634, and “HTML Destination” on page 634. For complete information about the ODS
destination statements, see theSAS Output Delivery System: User's Guide.

PLOTS= Option

Each statistical procedure that supports ODS Graphics has a PLOTS= option that you use to select graphs
and specify some options. The syntax of the PLOTS= option is as follows:

PLOTS < (global-plot-options) > < = plot-request< (options) > > ;

PLOTS < (global-plot-options) > < =(plot-request< (options) > < ... plot-request< (options) > >) > ;

The PLOTS= option has a common overall syntax for all statistical procedures, but the speci�cglobal-plot-
options, plot-requests, andplot-options vary across procedures. This section discusses only a few of the
options available in the PLOTS= option. For more information about the PLOTS= option, see the “Syntax”
section for each procedure that produces ODS Graphics. There are only a limited number of details that you
can control by using the PLOTS= option. Most graphical details are controlled either by graph templates (see
the section “Graph Templates” on page 722 in Chapter 22, “ODS Graphics Template Modi�cation”) or by
styles (see the section “ODS Styles” on page 643).

The PLOTS= option is usually speci�ed in the PROC statement. However, for some procedures, certain
analyses and hence certain plots can appear only if an additional statement is speci�ed. These procedures
might have a PLOTS= option in that other statement. For example, the PHREG procedure has a PLOTS=
option in the BAYES statement, which is used to perform a Bayesian analysis. For more information, see the
“Syntax” section of each procedure chapter. The following examples illustrate the syntax of the PLOTS=
option:

plots=all
plots=none
plots=residuals
plots=residuals(smooth)

plots=(trace autocorr)
plots(unpack)
plots(unpack)=diagnostics
plots=diagnostics(unpack)
plots(only)=freqplot
plots=(scree(unpack) loadings(plotref) preloadings(flip))
plots(unpack maxparmlabel=0 stepaxis=number)=coefficients
plots(sigonly)=(rawprob adjusted(unpack))

Also see the “Getting Started” sections “Survival Estimate Plot with PROC LIFETEST” on page 595,
“Contour and Surface Plots with PROC KDE” on page 596, “Contour Plots with PROC KRIGE2D” on
page 597, “LS-Means Diffogram with PROC GLIMMIX” on page 602, and “Principal Component Analysis
Plots with PROC PRINCOMP” on page 603 for examples of using the PLOTS= option.

The simplest PLOTS= speci�cations are of the form PLOTS=plot-request or PLOTS=(plot-requests). When
there is more than oneplot-request , theplot-request list must appear in parentheses. Eachplot-request either
requests a plot (for example, RESIDUALS) or provides a place to specify plot-speci�c options (for example,
DIAGNOSTICS(UNPACK)). Some simple and typicalplot-requests are explained next:

� PLOTS=ALL requests all plots that are relevant to the analysis. This does not mean that the procedure
produces all plots that it can produce. Plots that are produced for one set of options might not appear
when you specify PLOTS=ALL and a different set of options. In some cases, certain plots are not
produced unless certain options or statements outside the PLOTS= option are speci�ed.

� PLOTS=NONE disables ODS Graphics for just that step. You can use this option instead of specifying
ODS GRAPHICS OFF before a procedure step and ODS GRAPHICS ON after the step when you
want to suppress graphics for only that step.

� PLOTS=RESIDUALS requests a plot of residuals in a modeling procedure such as PROC REG.

� PLOTS=RESIDUALS(SMOOTH) requests the residuals plot along with a smooth �t function.

� PLOTS=(TRACE AUTOCORR) requests trace and autocorrelation plots in procedures that have
Bayesian analysis options.

Global-plot-options appear in parentheses after the option name and before the equal sign. These options
affect many or all of the plots. The UNPACK option is a commonly usedglobal-plot-option. It speci�es that
plots that are usually produced containing multiple plots per panel (or “packed”) should be unpacked so they
appear in multiple panels that each contain one plot. The speci�cation PLOTS(UNPACK)=(plot-requests)
unpacks all paneled plots. The UNPACK option is also used as an option in aplot-request when you want
to unpack only certain panels. For example, the option PLOTS=(DIAGNOSTICS(UNPACK) PARTIAL
PREDICTIONS) unpacks only the diagnostics panel. In some cases, unpacked plots contain additional
information that is not found in the smaller packed versions. The UNPACK option is not available for all
plot-requests; it is available only for plots that have multiple panels by default.

Another commonly usedglobal-plot-option is the ONLY option. Many procedures produce default plots, and
additional plots can be requested in the PLOTS= option. Specifying PLOTS=(plot-requests) while omitting
the default plots does not prevent the default plots from being produced. You use the ONLY option when
you want to see only the plots that are speci�cally cited in theplot-request list. Procedures that produce no
default plots usually do not provide an ONLY option. You can use ODS SELECT and ODS EXCLUDE
(see the section “Selecting and Excluding Graphs” on page 628) to select and exclude graphs, but in some
situations the ONLY option is more convenient. It is usually more ef�cient to select plots by using the

PLOTS(ONLY)= option, because the procedure does not do extra work to generate a plot that is excluded
by the PLOTS(ONLY)= option. In contrast, ODS SELECT and ODS EXCLUDE have their effect after the
procedure does the work to generate the plot.

Selecting and Viewing Graphs

This section describes techniques for selecting and viewing your graphs. Topics include the following:

� specifying an ODS destination for graphics
� viewing your graphs in the SAS windowing environment
� referring to graphs by name when using ODS
� selecting and excluding graphs from your output

Specifying an ODS Destination for Graphics

If you do not specify an ODS destination, then either the LISTING or HTML destination is used by default.
Here is an example of how you can explicitly specify the HTML destination:

ods graphics on;
ods html;

proc reg data=sashelp.class;
model Weight = Height;

run; quit;

ods html close;

This ODS HTML statement creates an HTML �le that has a default name. For information about specifying
a �lename, see the section “Specifying a File for ODS Output” on page 625. Other destinations are speci�ed
in a similar way. For example, you can specify an RTF destination by using the following statements:

ods graphics on;
ods rtf;

. . .

ods rtf close;

The destinations that ODS supports for graphics are as follows:

Destination Destination Family

DOCUMENT
HTML MARKUP
LATEX � MARKUP
LISTING
PCL PRINTER
PDF PRINTER
PS PRINTER
RTF

* The LATEX destination is experimental.

You can close all open destinations if you are interested only in displaying your output in a nondefault
destination. For example, if you want to see your output only in the RTF destination, you can specify the
following statements:

ods graphics on;
ods _all_ close;
ods rtf;

. . .

ods rtf close;
ods listing;

Closing unneeded destinations makes your jobs run faster and creates fewer �les. More generally, it makes
your jobs consume fewer resources, because a graph is created for every open destination. The last statement
opens the LISTING destination after you are �nished using the RTF destination.

You can also use the ODS OUTPUT destination to create an output data set from the data object that is used
to make a plot. Here is an example:

ods graphics on;

proc reg data=sashelp.class;
ods output fitplot=myfitplot;
model Weight = Height;

run; quit;

Specifying a File for ODS Output

You can specify a �lename for your output by using the FILE= option in the ODS destination statement, as in
the following example:

ods html file="test.htm";

The output is written to the �letest.htm, which is saved in the SAS current folder. At start-up, the SAS current
folder is the same directory in which you started your SAS session. If you are using the SAS windowing
environment, then the current folder is displayed in the status line at the bottom of the main SAS window.
If you do not specify a �lename for your output, then the SAS System provides a default �lename, which
depends on the ODS destination. This �le is saved in the SAS current folder. You can always check the
SAS log to verify the name of the �le in which your output is saved. For example, suppose you specify the
following statement:

ods html;

Then the following message is displayed in the SAS log:

NOTE: Writing HTML Body file: sashtml.htm

The default �lenames for each destination are speci�ed in the SAS Registry. For example, Figure 21.75
shows that the default �lename in the SAS Registry for the RTF destination issasrtf.rtf. For more information,
see theSAS Companionfor your operating system.

Viewing Your Graphs in the SAS Windowing Environment

The mechanism for viewing graphs that are created by ODS can vary depending on your operating system,
which viewers are installed on your computer, and the ODS destination that you select. If you do not specify
an ODS destination, then the default destination is either HTML or LISTING.

If you are using the SAS windowing environment and the HTML destination, then the results are displayed
by default in the SAS Results Viewer unless you are using an external browser. To use an external viewer,
selectTools I Options I Preferencesfrom the main SAS window. Then click theResultsandWeb tabs
to make your selection.

If you are using the PS destination, you must use a PostScript viewer, such as GSview. For information
about the windowing environment in a different operating system, see theSAS Companionfor that operating
system.

If you do not want to view the results as they are being generated, then selectToolsI Options I Preferences
from the main SAS window. Then on theResultstab, clear theView results as they are generatedcheck
box.

If you are using the SAS windowing environment and the LISTING destination, go to the Results window
and �nd the icon for the corresponding graph. You can double-click the graph icon to display the graph in the
default viewer that is con�gured on your computer for the corresponding image �le type (see Figure 21.14).

Determining Graph Names and Labels

Procedures assign a name to each graph that they create using ODS Graphics. This name enables you to
refer to ODS graphs in the same way that you refer to ODS tables (see the section “The ODS Statement” on
page 529 in Chapter 20, “Using the Output Delivery System”). You can determine the names of graphs in
several ways:

� You can look up graph names in the “ODS Graphics” section of chapters for procedures that use ODS
Graphics. For example, see the section “ODS Graphics” on page 7961 in Chapter 97, “The REG
Procedure.”

� You can use the Results window to view the names of ODS graphs that are created in your SAS session.
For more information, see the section “The SAS Results Window” on page 533 in Chapter 20, “Using
the Output Delivery System.”

� You can use the ODS TRACE ON statement to list the names of graphs that are created during your
SAS session. This statement adds identifying information in the SAS log (or optionally in the SAS
LISTING) for each graph that is produced. For more information, see the section “The ODS Statement”
on page 529 in Chapter 20, “Using the Output Delivery System.”

The graph name is not the same as the name of the image �le that contains the graph (see the section “Naming
Graphic Image Files” on page 631).

This example revisits the analysis that is described in the section “Contour and Surface Plots with PROC
KDE” on page 596. To determine which output objects are created by ODS, you specify the ODS TRACE
ON statement before the procedure statements as follows:

ods graphics on;
ods trace on;

proc kde data=bivnormal;
bivar x y / plots=contour surface;

run;

ods trace off;

The trace record from the SAS log is as follows:

Output Added:
-------------
Name: Inputs
Template: Stat.KDE.Inputs
Path: KDE.Bivar1.x_y.Inputs
-------------

Output Added:
-------------
Name: Controls
Template: Stat.KDE.Controls
Path: KDE.Bivar1.x_y.Controls
-------------

Output Added:
-------------
Name: ContourPlot
Label: Contour Plot
Template: Stat.KDE.Graphics.Contour
Path: KDE.Bivar1.x_y.ContourPlot
-------------

Output Added:
-------------
Name: SurfacePlot
Label: Density Surface
Template: Stat.KDE.Graphics.Surface
Path: KDE.Bivar1.x_y.SurfacePlot
-------------

By default, PROC KDE creates table objects named Inputs and Controls, and it creates graph objects named
ContourPlot and SurfacePlot. In addition to the name, the trace record provides the label, template, and path
for each output object. Graph templates are distinguished from table templates by a naming convention that
uses the procedure name in the second level andGraphics in the third level. For example, the fully quali�ed
template name for the surface plot that PROC KDE creates isStat.KDE.Graphics.SurfacePlot .

You can specify the LISTING option in the ODS TRACE ON statement to write the trace record to the
LISTING destination as follows:

ods trace on / listing;

Each table and graph has a path (or name path), which was previously shown in the trace output. The path
consists of the plot name, preceded by the names of one or more output groups. Each table and graph also
has a label path, which can be seen by adding the LABEL option to the ODS TRACE ON statement, after a
slash, as follows:

ods trace on / label;

proc kde data=bivnormal;
bivar x y / plots=contour surface;

run;

ods trace off;

A portion of the trace output is shown next:

Path: KDE.Bivar1.x_y.Inputs
Label Path: �The KDE Procedure�.�Bivariate Analysis�.�x and y�.�KDE.Bivar1.x_y�

Path: KDE.Bivar1.x_y.Controls
Label Path: �The KDE Procedure�.�Bivariate Analysis�.�x and y�.�KDE.Bivar1.x_y�

Path: KDE.Bivar1.x_y.ContourPlot
Label Path: �The KDE Procedure�.�Bivariate Analysis�.�x and y�.�Contour Plot�

Path: KDE.Bivar1.x_y.SurfacePlot
Label Path: �The KDE Procedure�.�Bivariate Analysis�.�x and y�.�Density Surface�

The label path contains the information that you see in the HTML table of contents. Names are �xed
(they do not vary and they are not data- or context-dependent). In contrast, labels often re�ect data- or
context-dependent information.

Selecting and Excluding Graphs

You can use the ODS SELECT and ODS EXCLUDE statements along with graph and table names to specify
which ODS outputs to display. For more information about how to use these statements, see the section “The
ODS Statement” on page 529 in Chapter 20, “Using the Output Delivery System.”

This section shows several examples of selecting and excluding graphs by using the data set and trace output
that are created in the section “Determining Graph Names and Labels” on page 626. The following statements



Selecting and Excluding Graphs F 629

use the ODS SELECT statement to select only two graphs, ContourPlot and SurfacePlot, for display in the
output:

proc kde data=bivnormal;
ods select ContourPlot SurfacePlot;
bivar x y / plots=contour surface;

run;

Equivalently, the following statements use the ODS EXCLUDE statement to exclude the two tables:

proc kde data=bivnormal;
ods exclude Inputs Controls;
bivar x y / plots=contour surface;

run;

You can select or exclude graphs by using either the name or the label. Labels must be speci�ed in quotes. In
the context of this example, the following two statements are equivalent:

ods select contourplot;
ods select �Contour Plot�;

You can also specify multiple levels of the path, as in the following examples:

ods select x_y.contourplot;
ods select �x and y�.�Contour Plot�;
ods select �x and y�.contourplot;
ods select x_y.�Contour Plot�;

You can mix name and label paths, as in the last two statements. All four of the preceding statements
select the same plot. Furthermore, selection based directly on the names and labels is case-insensitive. The
following statements all select the same plot:

ods select x_y.contourplot;
ods select �x and y�.�Contour Plot�;
ods select X_Y.CONTOURPLOT;
ods select �X AND Y�.�CONTOUR PLOT�;

It is sometimes useful to specify a WHERE clause in an ODS SELECT or ODS EXCLUDE statement.
This enables you to specify expressions based on either the name path or the label path. You can base your
selection on two automatic variables,_path_ and_label_. The following two statements select every object
whose path contains the string `plot' and every object whose label path contains the string `plot', respectively,
ignoring the case in the name and label:

ods select where = (lowcase(_path_) ? �plot�);
ods select where = (lowcase(_label_) ? �plot�);

The question mark operator means that the second expression (the string `plot') is contained in the �rst
expression (the lowercase version of the name or label). For example, all the following names match `plot' in
the WHERE clause: plot, SurfacePlot, SURFACEPLOT, FitPlot, pLoTtInG, Splotch, and so on. Because
WHERE clause selection is based on SAS string comparisons, selection is case-sensitive. The LOWCASE
function is used to ensure a match even when the case of the speci�ed string does not match the case of the
actual name or label.

WHERE clauses are particularly useful when you want to select all the objects in a group. A group is a level
of the name path or label path hierarchy before the last level. In the following step, all the objects whose

name path contains `DiagnosticPlots' are selected:

proc reg data=sashelp.class plots(unpack);
ods select where = (_path_ ? �DiagnosticPlots�);
model Weight = Height;

run; quit;

These are the plots that come from unpacking the PROC REG diagnostics panel of plots. All are in a group
named `DiagnosticPlots'.

Graphic Image Files

Accessing your graphs as individual image �les is useful when you want to include them in various types of
documents. The default image �le type depends on the ODS destination, but you can specify other supported
image �le types. You can also specify the names for your graphic image �les and the directory in which you
want to save them. This section describes the image �le types that ODS Graphics supports, and it explains
how to name and save graphic image �les.

Image File Types

If you are using the LISTING or HTML destination, your graphs are individually produced in a speci�c
image �le type, such as PNG. If you are using a destination in the PRINTER family or the RTF destination,
the graphs are contained in the ODS output �le and cannot be accessed as individual image �les. However,
you can open an RTF output �le in Microsoft Word and then copy and paste the graphs into another document,
such as a Microsoft PowerPoint presentation. This is illustrated in Example 21.2.

Table 21.2 shows the various ODS destinations that ODS Graphics supports, the viewer that is appropriate
for displaying graphs in each destination, and the image �le types that each destination supports.

Table 21.2 Destinations and Image File Types Supported by ODS Graphics

Destination
Destination
Family Recommended Viewer Image File Types

DOCUMENT Not applicable Not applicable
HTML MARKUP Web browser PNG (default), GIF, JPEG
LATEX MARKUP PostScript or PDF viewer after the LATEX

�le is compiled
PostScript (default), EPSI,
GIF, JPEG, PDF, PNG

LISTING Default viewer in your system for the
speci�ed �le type

PNG (default), BMP, DIB,
EMF, EPSI, GIF, JFIF, JPEG,
PBM, PS, TIFF, WMF

PCL PRINTER Not applicable Contained in PRN �le
PDF PRINTER PDF viewer, such as Adobe Reader Contained in PDF �le
PS PRINTER PostScript viewer, such as GSview Contained in PostScript �le
RTF Word processor, such as Microsoft Word Contained in RTF �le

For destinations such as PDF and RTF, you can control the types of the images that the �le contains even
though individual �les are not made for each image. The default image �le type is PNG, and other image
types are available. For more information, see theSAS Output Delivery System: User's Guide.

Scalable Vector Graphics

Scalable vector graphics output is supported in ODS Graphics. The output type support depends on the ODS
destination that you use. You can specify the OUTPUTFMT= option in the ODS GRAPHICS statement to
specify the output type for any destination. For destinations that generate vector graphics by default, you can
get image output by specifying OUTPUTFMT=STATIC.

Vector graphics are not supported for all graph types. When vector graphics are requested but not supported,
the graph automatically changes to image output. Vector graphics are not supported for the following graph
types:

� three-dimensional graph
� contour plots that have smooth gradient �lls
� graphs that have continuous legends
� graphs that have data skins (such as bars charts that appear to be shiny or have depth)
� graphs that have rotated annotation images
� graphs that have transparency (EMF and PS only)

The LISTING destination can generate all the supported forms of vector-based output: PDF, PS, EMF, SVG,
and PCL. Each graph is generated in a separate �le that can be included in a larger report. The default output
format is a PNG image.

Like the LISTING destination, the ODS PRINTER destination can generate all the supported vector output
types. The output format depends on the type of printer that you select. If you select the PDF, SVG, or
PCL5C printer, vector-based output is automatically produced. However, if you select the PS or EMF printer,
you need to set the OUTPUTFMT= option in the ODS GRAPHICS statement to PS or EMF, respectively, to
create vector-based output. By default, the output from this destination is contained in one �le instead of
individual �les for each graph.

The PDF destination produces PDF vector output by default, except for the exceptions noted. You can specify
OUTPUTFMT=STATIC in the ODS GRAPHICS statement to produce an embedded image in the PDF �le.

The experimental LATEX destination produces PS vector output by default, except for the exceptions noted.
You can specify OUTPUTFMT=STATIC in the ODS GRAPHICS statement to produce an embedded image
in the PostScript �le.

The RTF destination produces PNG image output by default. It also supports vector-based EMF output for
this destination. You can specify OUTPUTFMT=EMF in the ODS GRAPHICS statement to select this output
type. If one of the noted exceptions occurs, the output type for that graph changes to a PNG image type.

The HTML destination produces PNG image output by default. It also supports vector-based SVG output for
this destination. You can specify OUTPUTFMT=SVG in the ODS GRAPHICS statement to select this output
type. If one of the noted exceptions occurs, the output type for that graph changes to a PNG image type.

In most cases, the vector graphics �le is much smaller than a comparable static image �le. However, in some
cases, the vector graphics �le is larger than the image version. This is likely for scatter plots of data sets that
have a large number of observations.

Naming Graphic Image Files

The following discussion applies to the destinations where ODS graphs are created as individual image �les
(for example, HTML and LISTING). The names of graphic image �les are determined by a base �lename,

an index counter, and an extension. By default, the base �lename is the ODS graph name (see the section
“Determining Graph Names and Labels” on page 626). There is an index counter for each base �lename.
The extension indicates the image �le type. The �rst time a graph object that has a particular base �lename
is created, the �lename consists only of the base �lename and the extension. If a graph that has the same
base �lename is created multiple times, then an index counter is appended to the base �lename to avoid
overwriting previously created images.

To illustrate, consider the following statements:

proc kde data=bivnormal;
ods select ContourPlot SurfacePlot;
bivar x y / plots=contour surface;

run;

If you run this step at the beginning of a SAS session, the two graphic image �les that are created are
ContourPlot.png andSurfacePlot.png. If you immediately rerun these statements, then ODS creates the same
graphs in different image �les namedContourPlot1.png andSurfacePlot1.png. The next time, the image �les
are namedContourPlot2.png andSurfacePlot2.png. The index starts at 0, and 1 is added each time the same
name is used. However, if the index is at 0, then 0 is not included in the �lename.

Resetting the Index Counter

You can specify RESET=INDEX in the ODS GRAPHICS statement to reset the index counter. This is useful
when you need to have predictable names. It is particularly useful when you are running a SAS program
multiple times in the same session. The following statement resets the index:

ods graphics on / reset=index;

The index counter is reinitialized at the beginning of your SAS session or if you specify RESET=INDEX in
the ODS GRAPHICS statement. Graphic image �les that have the same name are overwritten.

Specifying Base Filenames

You can specify a base �lename for all your graphic image �les by using the IMAGENAME= option in the
ODS GRAPHICS statement as follows:

ods graphics on / imagename="MyName";

You can also specify RESET=INDEX as follows:

ods graphics on / reset=index imagename="MyName";

The IMAGENAME= option overrides the default base �lename. In the preceding statement, the graphic
image �les are namedMyName, MyName1, MyName2, and so on.

Specifying Image File Types

You can specify the image �le type for the LISTING, HTML, or LATEX destination by specifying OUT-
PUTFMT= in the ODS GRAPHICS statement as follows:

ods graphics on / outputfmts=gif;

For more information, see the section “ODS GRAPHICS Statement” on page 618.

Naming Graphic Image Files with Multiple Destinations

Because the index counter depends only on the base �lename, if you specify multiple ODS destinations for
your output, then the index counter is increased independently of the destination. For example, the following
statements create image �les namedContourPlot.png andSurfacePlot.png that correspond to the LISTING
destination andContourPlot1.png andSurfacePlot1.png that correspond to the HTML destination:

ods listing;
ods html;
ods graphics on / reset;

proc kde data=bivnormal;
ods select ContourPlot SurfacePlot;
bivar x y / plots=contour surface;

run;

ods _all_ close;
ods listing;

When you specify one of the destinations in the PRINTER family or the RTF destination, your ODS graphs
are embedded in the document, so the index counter is not affected. For example, the following statements
create the image �lesContourPlot.png andSurfacePlot.png for the LISTING destinations but no image �les for
the RTF destination:

ods listing;
ods rtf;
ods graphics on / reset;

proc kde data=bivnormal;
ods select ContourPlot SurfacePlot;
bivar x y / plots=contour surface;

run;

ods _all_ close;

Saving Graphic Image Files

Knowing where your graphic image �les are saved and how they are named is particularly important if you
are running in batch mode, if you have disabled the SAS Results window (see the section “Viewing Your
Graphs in the SAS Windowing Environment” on page 626), or if you plan to access the �les for inclusion in a
paper or presentation. The following discussion assumes that you are running SAS on the Windows operating
system. If you are running on a different operating system, see theSAS Companionfor your operating system.

In the SAS windowing environment, the current folder is displayed in the status line at the bottom of the
main SAS window. When theUse WORK folder check box is cleared on theResultstab (which you access
by selectingTools I Options I Preferencesfrom the main SAS window), graphic image �les are saved in
the current folder and available after your SAS session ends. They can accumulate over time and take up a
great deal of space. WhenUse WORK folder is selected, graphic image �les are stored in theWork folder
and are not available after your SAS session ends.

If you are running your SAS programs in batch mode, the graphs are saved by default in the same directory
where you started your SAS session. For example, suppose the SAS current folder isC:\my�les . If ODS

Graphics is enabled, then your graphic image �les are saved in the directoryC:\my�les . Traditional graphics
are always saved in a catalog in yourWork directory.

If you are using the LISTING, HTML, and LATEX destinations, you can specify a directory for saving your
graphic image �les. If you are using a destination in the PRINTER family or the RTF destination, you can
specify a directory only for your output �le. The remainder of this discussion provides details about each
destination type.

LISTING Destination

If you are using the LISTING destination, the individual graphs are created as PNG �les by default. You can
use the GPATH= option in the ODS LISTING statement to specify the directory where your graphics �les
are saved. For example, if you want to save your graphic image �les in theC:\�gures directory, then you can
specify the following:

ods listing gpath="C:\figures";

It is important to note that the GPATH= option applies only to ODS Graphics. It does not affect the behavior
of graphs that are created by traditional SAS/GRAPH procedures.

HTML Destination

If you are using the HTML destination, the individual graphs are created as PNG �les by default. You can use
the PATH= and GPATH= options in the ODS HTML statement to specify the directory where your HTML
and graphics �les, respectively, are saved. These options also give you more control over your graphs. For
example, if you want to save your HTML �le namedtest.htm in theC:\my�les directory, but you want to save
your graphic image �les inC:\my�les\png , then you can specify the following:

ods html path = "C:\myfiles"
gpath = "C:\myfiles\png"
file = "test.htm";

When you specify the URL= suboption along with the GPATH= option, SAS creates relative paths for the
links and references to the graphic image �les in the HTML �le. This is useful for building output �les that
are easily moved from one location to another. For example, the following statements create a relative path to
thepng directory in all the links and references that the �letest.htm contains:

ods html path = "C:\myfiles"
gpath = "C:\myfiles\png" (url="png/")
file = "test.htm";

If you do not specify the URL= suboption, SAS creates absolute paths that are hard-coded in the HTML �le.
These can cause broken links if you move the �les. For more information, see the ODS HTML statement in
theSAS Output Delivery System: User's Guide.

LATEX Destination (Experimental)

LATEX is a document preparation system for high-quality typesetting of mathematical and scienti�c material.
The ODS LATEX statement produces output in the form of a LATEX source �le that is ready to compile in
LATEX. When you request ODS Graphics for a LATEX destination,2 ODS creates the requested graphs as
PostScript �les by default, and the LATEX source �le includes references to these graphic image �les. You can
compile the LATEX �le, or you can ignore this �le and simply access the individual PostScript �les to include

2The LATEX destination is experimental.

your graphs in a different LATEX document, such as a paper that you are writing. You can specify the PATH=
and GPATH= options in the ODS LATEX statement, as explained previously for the ODS HTML statement;
see Example 21.3 for an illustration. The ODS LATEX statement is an alias for the ODS MARKUP statement
with the TAGSET=LATEX option. For more information, see theSAS Output Delivery System: User's Guide.

The default image �le type for the LATEX destination is PostScript. When you use LATEX to compile your
document, the graphics format for included images is PostScript. However, if you prefer to use pdfLATEX,
you can specify a different format, such as JPEG, PDF, or PNG, any of which can be directly included in
your pdfLATEX document. To specify one of these formats, you use the OUTPUTFMT= option in the ODS
GRAPHICS statement. For more information, see the LATEX documentation for thegraphicx package.

Creating Graphs in Multiple Destinations

This section illustrates how to send your output to more than one destination by using a single execution
of your SAS statements. For example, to create LISTING, HTML, and RTF output, you can specify the
ODS LISTING, ODS HTML, and ODS RTF statements before your procedure statements. The ODS _ALL_
CLOSE statement closes all open destinations before and after the other statements are run.

ods _all_ close;
ods listing;
ods html;
ods rtf;

. . .

ods _all_ close;

You can also specify multiple instances of the same destination. For example, using the data in the section
“Contour and Surface Plots with PROC KDE” on page 596, the following statements save the contour plot to
the �le contour.pdf and the surface plot to the �lesurface.pdf :

ods _all_ close;
ods pdf file="contour.pdf";
ods pdf select ContourPlot;
ods pdf(id=srf) file="surface.pdf";
ods pdf(id=srf) select SurfacePlot;
ods graphics on;

proc kde data=bivnormal;
ods select ContourPlot SurfacePlot;
bivar x y / plots=contour surface;

run;

ods _all_ close;

The ID= option assigns the namesrf to the second instance of the PDF destination. Without the ID= option,
the second ODS PDF statement closes the destination that was opened by the �rst ODS PDF statement, and
it opens a new instance of the PDF destination. In that case, the �lecontour.pdf is not created. For more
information, see the ODS PDF statement in theSAS Output Delivery System: User's Guide.

Graph Size and Resolution

ODS provides options for specifying the size and resolution of graphs. You can specify the size of a graph in
the ODS GRAPHICS statement and the resolution in an ODS destination statement. There are two other
ways to change the size of a graph, but they are rarely needed. The three methods are as follows:

� Usually, you specify the WIDTH= or HEIGHT= option (or both) in the ODS GRAPHICS statement to
change the size of a graph.

� You can modify the size of a particular graph by specifying the dimensions in the DESIGNHEIGHT=
and DESIGNWIDTH= options in the BEGINGRAPH statement in the template. Some templates
contain the speci�cation DESIGNWIDTH=DEFAULTDESIGNHEIGHT, which sets the width of the
graph to the default height, or DESIGNHEIGHT=DEFAULTDESIGNWIDTH, which sets the height
of the graph to the default width.

� You can modify the size of all your ODS graphs by specifying the dimensions in the OUT-
PUTHEIGHT= and OUTPUTWIDTH= options in the style template.

The following examples show typical ways to change the size of your graphs:

ods graphics on / width=6in;
ods graphics on / height=4in;
ods graphics on / width=4.5in height=3.5in;

The dimensions of the graph can be speci�ed in pixels (such as 200PX), inches (such as 3IN), or centimeters
(such as 8CM). The default dimensions of ODS Graphics are 640 pixels wide and 480 pixels high, and these
values determine the default aspect ratio. The actual size of the graph in inches depends on your printer or
display device. For example, if the resolution of your printer is 100 dots per inch (DPI) and you want a graph
that is 4 inches wide, you should set the width to 400 pixels.

If you specify only one dimension, the other dimension is determined by the default aspect ratio—that is,
height = 0.75� width. For best results, you should create your graphs by using the exact size that is used to
display the graphs in your paper or presentation. In other words, avoid generating graphs at one size and then
expanding or shrinking them for inclusion in your document.

By default, fonts and symbol markers are automatically scaled along with the size of the graph. You can
suppress this scaling by specifying the SCALE= option, as in the following example:

ods graphics on / scale=off;

The default resolution of graphs that are created in the HTML and LISTING destinations is 96 DPI, whereas
the default for the RTF destination is 200 DPI. The 200 DPI value is recommended if you are copying
and pasting graphs into a Microsoft PowerPoint presentation or a Microsoft Word document. Graphs in
SAS/STAT documentation are usually generated at 300 DPI for display in PDF and 96 DPI for display in
HTML.

You can change the resolution by using the IMAGE_DPI= option in any ODS destination statement, as in the
following example:

ods html image_dpi=300;

An increase in resolution often improves the quality of the graphs, but it also greatly increases the size of
the image �le. Going from 96 DPI to 300 DPI increases the size of the image �le by approximately a factor
of .300=96/2 D 9:77. Even when you are using a higher resolution for most of your graphs, you should
consider using a lower resolution for some, such as contour plots, that create large �les even at a lower
resolution.

If you increase the resolution, you might need to compensate by reducing the size of the graph, as in the
following example:

ods graphics on / width=4.5in height=3.5in;

Increasing resolution also increases the amount of memory that is needed for your program to run. You can
increase the amount of memory available to ODS Graphics by specifying an option when you invoke SAS, as
in the following example:

-jreoptions �(-Xmx256m)�

You can modify the default amount of memory available to ODS Graphics by changing JREOPTIONS in
your SAS con�guration �le to the settings–Xmxnnnm –Xmsnnnm, wherennn is the amount of memory in
megabytes. An example is–Xmx256m –Xms256m. In either case, the exact syntax varies depending on
your operating system, and the amount of memory that you can allocate varies from system to system. For
more information, see theSAS Companionfor your operating system.

ODS Graphics Editor

The ODS Graphics Editor is a point-and-click interface that you can use to modify a speci�c graph created
by ODS Graphics. For example, if you need to enhance a graph for a paper or presentation, you can use
the ODS Graphics Editor to customize the title, modify the axis labels, annotate particular data points, and
change graph element properties such as fonts, colors, and line styles.

This section explains how to enable ODS Graphics to create editable graphs and how to invoke the ODS
Graphics Editor. You can use the ODS Graphics Editor in the SAS windowing environment, provided that the
LISTING destination is open and you have �rst enabled ODS Graphics to create editable graphs.NOTE : The
LISTING destination is usually open by default. There are three steps that you must take to edit a graph:

1 First enable the creation of editable graphs in one of three ways:
� Use an ODS statement to temporarily enable this feature.
� Use a SAS command to temporarily enable this feature.
� Use the SAS Registry Editor to permanently enable this feature.

Creating editable graphs requires additional resources, so you might not want to permanently enable this
feature.

2 Submit your SAS code and create editable graphs.

3 Invoke the ODS Graphics Editor and edit the plot.

Step 2 involves submitting SAS code in the usual way; no special instructions are needed for creating graphs
that can be edited. Steps 1 and 3 are explained in more detail in the following sections.

Enabling the Creation of Editable Graphs

Temporarily Enable Creation of Editable Graphs by Using an ODS Statement

You can enable the creation of editable graphs within a SAS session by submitting one of the following
statements:

ods listing sge=on;
ods html sge=on;

You can disable the creation of editable graphs by submitting one of the following statements:

ods listing sge=off;
ods html sge=off;

Permanently Enable Creation of Editable Graphs across SAS Sessions

You can create a default setting that enables or disables the creation of editable graphs across SAS sessions
by using the “ODS Graphics Editor” setting in the SAS Registry. You can change this setting in the SAS
windowing environment as follows:

1 Open the Registry Editor by enteringregedit on the command line.

2 SelectSAS_REGISTRYI ODSI GUI I RESULTS.

3 Click ODS Graphics Editor to open the Edit String Value window. In theValue Data �eld, type On to
enable the creation of editable graphs orOff to disable it.

4 Click OK.

Editing a Graph with the ODS Graphics Editor

The ODS Graphics Editor is illustrated using the following example:

data growth;
length Country $ 20;
input country &$ GDP LFG EQP NEQ GAP;
datalines;

Argentina 0.0089 0.0118 0.0214 0.2286 0.6079
Austria 0.0332 0.0014 0.0991 0.1349 0.5809

... more lines ...

Zambia -0.0110 0.0275 0.0702 0.2012 0.8695
Zimbabwe 0.0110 0.0309 0.0843 0.1257 0.8875
;

ods graphics on;
ods html sge=on;

proc robustreg data=growth plots=(ddplot histogram);
model GDP = LFG GAP EQP NEQ / diagnostics leverage;
output out=robout r=resid sr=stdres;

run;

ods _all_ close;
ods listing;

The DATA and PROC ROBUSTREG steps are submitted to the SAS System, in this case from the SAS
windowing environment, as shown in Figure 21.17. Two versions of the graph are created: one in an
uneditable PNG �le (for example,DDPlot.png) and one in an editable SGE �le (for example,DDPlot.sge).
Both are saved in the SAS current folder. You can edit the graph in one of three ways:

� In the Results window, double-click the second graph icon for the graph that you want to edit (see
Figure 21.17). The second graph icon corresponds to the SGE �le, and the �rst graph icon corresponds
to the PNG �le. Clicking the �rst graph icon invokes a host-dependent graph viewer (for example,
Microsoft Photo Editor in Windows), not the ODS Graphics Editor.NOTE : The ODS Graphics Editor
window might be hidden behind other windows in the SAS windowing environment.

� You can edit the graph by selecting it in the SAS Explorer window. You must �rst navigate to the SAS
current folder and to the SGE �les.

� You can open the graph from outside the SAS System. For example, if you are running SAS on the
Windows operating system, you can click on the graph's SGE �le to open it by using the ODS Graphics
Editor.

